/* The SAS program (HEI-2010 at the population/group level using NHANES 2003-2004 data (and MPED))
Population Ratio Method.SAS*/

/*INSTRUCTIONS – complete tasks 1-4 in this section, and run these SAS codes before proceeding to the HEI-2010 scoring program that follows*/

/*1. Create a folder on your computer “home folder”, and save the FPED data, NHANES data, Demographic data, and the required HEI-2010 macros in it. Specify the path to the folder. */

%let home = C:\Users\Documents\MPED_NHANES; /*In this Example, the “home” folder is in C Drive, within Documents, and is called MPED_NHANES */
/*2. Libnames here specify the input files. */

libname NH “&home\NH”; 
libname MPED “&home\MPED”; /*In this Example, the MPED data are in a folder called “MPED”, and the NHANES and Demographic data are in a folder called “NH”, all saved within the “home” folder. These are SAS datasets. */
/*3. Create a folder in the "Home" folder, where the output file, containing HEI-2010 component and total scores are to be exported. Specify the name of the folder. */

filename RES “&home\RES”; /*In this Example, the folder is called “RES”, within the “home” folder, and the exported results will be a csv file called “result”. */
/*4. Read in required HEI-2010 scoring macros. These macros must be saved within the home folder. */
%include “&home\hei2010.beanspeas.allocation.macro.sas”;
%include “&home\hei2010.score.macro.sas”;
/*NOTE: Once you have completed all the steps above, all you need to do is run the SAS program below. Unless you used different names for your folders and datasets, no other action is required from you. */
title 'HEI-2010 scores for NHANES 2003-2004 day 1, AGE >= 2, RELIABLE DIETS, Include Pregnant and Lactating Women';

/*Section I. Calculations at the individual participant level to obtain variables needed to calculate HEI-2010 scores.*/
/*Step 1: Locate the required datasets and variables and make necessary edits to the datasets */
*Part a: Read in the MPED, which gives My Pyramid Equivalents per 100 grams of food, and perform edits and corrections;
data MPED;

 set MPED.EQUIV0304;

/*rename variables;  */
  FOODCODE=1*DRDIFDCD;

  MODCODE=1*DRDIMC; 

/*Replace some food codes with values to correct for previously identified errors in the database.

NOTE: This correction is specific to 2003-2004 NHANES data.  This change should be commented out for any other year of NHANES data.*/ 

*pizza (n=3) values from FPED 11/12 data;
if FOODCODE = 58106210 then do;

   G_TOTAL=1.88;

   G_WHL=0;

   G_NWHL=1.88;

   V_TOTAL=0.12;

   V_TOMATO=0.12;

   D_TOTAL=0.70;

   D_CHEESE=0.70;

   DISCFAT_OIL=0.44;

   DISCFAT_SOL=8.00;

   ADD_SUG=0.19;

   end;

else if FOODCODE = 58106220 then do;

   G_TOTAL=1.75;

   G_WHL=0;

   G_NWHL=1.75;

   V_TOTAL=0.12;

   V_TOMATO=0.12;

   D_TOTAL=0.66;

   D_CHEESE=0.66;

   DISCFAT_OIL=0.44;

   DISCFAT_SOL=10.62;

   ADD_SUG=0.19;

   end;

else if FOODCODE = 58106230 then do;

   G_TOTAL=1.88;

   G_WHL=0;

   G_NWHL=1.88;

   V_TOTAL=0.12;

   V_TOMATO=0.12;

   D_TOTAL=0.66;

   D_CHEESE=0.66;

   DISCFAT_OIL=0.44;

   DISCFAT_SOL=8.82;

   ADD_SUG=0.19;

   end;

/* Move soy beverages out of soybean products into dairy; */
/*In the HEI-2010, soy beverages are counted as part of the Milk component. Convert the four soy beverage codes in the MPED from M_SOY oz equivalents to D_TOTAL cup equivalents using the following conversion process*/
/*FOODCODE=11310000, MILK, IMITATION, FLUID, SOY BASED (1 cup=244 grams)

FOODCODE=11320000, MILK, SOY, READY-TO-DRINK, NOT BABY (1 cup=245 grams)

FOODCODE=11321000, MILK, SOY, READY-TO-DRINK, NOT BABY'S, CHOCOLATE (1 cup=240 grams)

FOODCODE=11330000, MILK, SOY, DRY, RECONSTITUTED, NOT BABY (1 cup=245 grams)*/
if FOODCODE=11310000 then do;

     M_SOY=0;


 D_TOTAL=round(100*(1/244),.001);


 end;

  else if FOODCODE=11320000 then do;

     M_SOY=0;


 D_TOTAL=round (100*(1/245),.001);


 end;

  else if FOODCODE=11321000 then do;

     M_SOY=0;


 D_TOTAL=round (100*(1/240),.001);


 end;

  else if FOODCODE=11330000 then do;

     M_SOY=0;


 D_TOTAL=round (100*(1/245),.001);


 end;

run;

*part b: Read in the CNPP MyPyramid Equivalents Databases for Whole Fruit and Fruit Juice for NHANES 2003-04;
data CNPPMPED_WJFRT;

  set MPED.cnppmyp_v1nhanes0304_wjfrt; 

run;

*part c: Read in the NHANES 2003-2004 Individual Foods, First Day dataset for people with satisfactory recall status;
data FOOD; 

  set NH.DR1IFF_C;

  FOODCODE=1*DR1IFDCD; /*convert variable name and type*/
  MODCODE=1*DR1MC;

  if DR1DRSTZ=1; /*reliable dietary recall status*/
run;

*part d: Read in the NHANES 2003-2004 Total Nutrient Intakes, First Day dataset if reliable recall status;
data NUTRIENT (keep=SEQN WTDRD1 DR1TKCAL DR1TSFAT DR1TALCO DR1TSODI DR1DRSTZ DR1TMFAT DR1TPFAT);

  set NH.DR1TOT_C;

  if DR1DRSTZ=1; /*reliable dietary recall status*/
run;

*part e: Read in the NHANES 2003-2004 Demographic dataset for persons aged two and older;
data DEMO (keep=SEQN RIDAGEYR RIAGENDR SDDSRVYR SDMVPSU SDMVSTRA);

  set NH.DEMO_C;

  if RIDAGEYR >= 2;

run;

/*Step 2: Combine the required datasets*/
*part a: combine MPED and WHOLE FRUIT data on a food level;
proc sort data=MPED;

  by FOODCODE MODCODE;

run;

proc sort data=CNPPMPED_WJFRT;

  by FOODCODE MODCODE;

run;

data NEWMPED;

  merge MPED CNPPMPED_WJFRT;

  by FOODCODE MODCODE;

run;

*part b: combine food intake and MPED plus WHOLE FRUIT data on a food level;
proc sort data=FOOD;

  by FOODCODE MODCODE;

run;

data FDPYR;

  merge FOOD (in=S) NEWMPED (in=N);

  by FOODCODE MODCODE;

  if S and N;

run;

*part c: convert individuals' food intake amounts from grams to 

MyPyramid equivalents;
data FDPYR;

  set FDPYR;

 array PYRVAR G_TOTAL--A_BEV WHOLEFRT FRTJUICE;

 do over PYRVAR;

    PYRVAR=PYRVAR*(DR1IGRMS/100);

 end;

run;

*part d: calculate individual food intake amounts for MyPyramid food groups for one day;
proc sort data=FDPYR;

  by SEQN;

run;

proc means data=FDPYR noprint;

  by SEQN;

  var G_TOTAL--A_BEV WHOLEFRT FRTJUICE;

  output out=PYRCALC sum= ;

run;

*part e: combine nutrient and demographic data on a person level;
proc sort data=NUTRIENT;

  by SEQN;

run;

proc sort data=DEMO;

  by SEQN;

run;

data COHORT;

  merge NUTRIENT (in=N) DEMO (in=D);

  by SEQN;

  if N and D;

run;

*part f: combine all data on a person level;
data BOTH;

  merge COHORT (in=C) PYRCALC(in=P);

  by SEQN;

  if C;   

run;

/*Please note: MyPyramid equivalent values for total vegetable intake (V_TOTAL) in the HEI-2010may be different from V_TOTAL in the MPED because legumes may be counted as vegetables or meat in the HEI-2010; and total dairy intake (D_TOTAL) in the HEI-2010 may be different from D_TOTAL in the MPED because soy beverages are counted as milk in the HEI-2010.*/
*part g: set pryvar to zero for people with zero cal who do not appear in iff data;
data BOTH;

  set BOTH;

  by SEQN;

 array PYRVAR G_TOTAL--A_BEV WHOLEFRT FRTJUICE;

 do over PYRVAR;

    if PYRVAR < 0 then PYRVAR =0;

 end;

run;

/*Step 3: Creates additional required variables: MONOPOLY, ALLMEAT, SEAPLANT, ADDSUGC, SOLFATC, EXALCCAL, and EMPTYCAL10. */
data BOTH;

  set BOTH;

  by SEQN;

  MONOPOLY=DR1TMFAT+DR1TPFAT;

  ALLMEAT=M_MPF+M_EGG+M_NUTSD+M_SOY;

  SEAPLANT=M_FISH_HI+M_FISH_LO+M_SOY+M_NUTSD;

 /**Calculate intake of Empty Calories**/
      ADDSUGC=16*ADD_SUG;
/*calories from added sugars*/
      SOLFATC=DISCFAT_SOL*9;
/*calories from solid fat*/
      MAXALCGR=13*(DR1TKCAL/1000);  /*max grams of alcohol based on kcal intake*/
      if DR1TALCO <= MAXALCGR then EXALCCAL=0;  /*consumed less than max*/
        else if DR1TALCO > MAXALCGR then EXALCCAL=7*( DR1TALCO - MAXALCGR);  /*get cal from extra alc grams*/
      EMPTYCAL10=ADDSUGC+SOLFATC+EXALCCAL; /*total empty calories in hei2010 definition*/
run; 

/*Section (II): Calculation of weighted means and a variance-covariance matrix and generation of a Monte Carlo dataset, enabling standard errors to be calculated.*/
/*Step 1.  Calculate the weighted means and the variance/covariance matrix for the dietary variables of interest. */
data ONE;

    set BOTH;

    array comp (15) DR1TKCAL V_TOTAL V_DRKGR LEGUMES F_TOTAL WHOLEFRT G_WHL D_TOTAL 

      ALLMEAT SEAPLANT MONOPOLY DR1TSFAT DR1TSODI G_NWHL EMPTYCAL10;

    * Turn each variable into an observation of the single variable VBL ;
    * keeping track of the order with the dum_num variable; 

    do i = 1 to 15;

        VBL = comp(i);

        dum_num = i;

        output;

        end;

run;

data ONE;

    set ONE;

    * Create dummies with the same names as the original variables;
    * The i-th dummy gets a 1 if the observation is associated with the

    * i-th original variable;
    array comp (15) DR1TKCAL V_TOTAL V_DRKGR LEGUMES F_TOTAL WHOLEFRT G_WHL D_TOTAL 

      ALLMEAT SEAPLANT MONOPOLY DR1TSFAT DR1TSODI G_NWHL EMPTYCAL10;

    do i = 1 to 15;

        if dum_num = i then comp(i) = 1;

        else comp(i) = 0;

        end;

    drop i dum_num;

run;

* Now run PROC (SURVEY)REG with dependent variable VBL and all the dummies;
* as predictors. Force the regression to not include an intercept, and ask;
* for the covariance matrix of the parameter estimates. ;
proc surveyreg data=ONE;

    strata SDMVSTRA;

    cluster SDMVPSU;

    weight WTDRD1;

    model VBL= DR1TKCAL V_TOTAL V_DRKGR LEGUMES F_TOTAL WHOLEFRT G_WHL D_TOTAL 

      ALLMEAT SEAPLANT MONOPOLY DR1TSFAT DR1TSODI G_NWHL 

EMPTYCAL10/noint covb;

    * Output the covariance matrix we wanted all along;
    ods output COVB=CSD_COV;

    title2 "Tricking SURVEYREG into giving us the covariance matrix of means";

run;

* this proc print is for verification and is not necessary;
* uncomment next three lines if more information is desired;
* proc print data= CSD_COV;
*    title2 "Printout of csd_cov dataset -uses complex survey info";
*run;
*If curious about results then remove the noprint option and more ;
* output will be displayed;
proc means data=BOTH n min max mean noprint;

  weight WTDRD1;

  var DR1TKCAL V_TOTAL V_DRKGR LEGUMES F_TOTAL WHOLEFRT G_WHL D_TOTAL 

      ALLMEAT SEAPLANT MONOPOLY DR1TSFAT DR1TSODI G_NWHL EMPTYCAL10;

  title2 'look at weighted means';

  output out=WTDM mean= ;

run;

data COVDATA (drop=Parameter);

  set CSD_COV;

  _TYPE_='COV ';

  _NAME_=Parameter;

run;

data WTDM (drop=_TYPE_ _FREQ_);

  set WTDM;

run;

data WTDM;

  set WTDM;

  _TYPE_='MEAN';

run;

data COVDATA;

  set COVDATA WTDM;

run;

* this proc print is for verification and is not necessary;
* uncomment next three lines if more information is desired;
* proc print data=COVDATA;
*  title2 'input to simnorml';
*run;
/*Step 2.  In this step, a Monte Carlo data set with 10,000 rows is generated using the means and variance/covariance matrix from step 1*/
*seed value may be changed using a random number generator;
proc simnormal data=COVDATA(type=cov) numreal=10000 seed=51230077 outseed out=SIM_DATA;

  var DR1TKCAL V_TOTAL V_DRKGR LEGUMES F_TOTAL WHOLEFRT G_WHL D_TOTAL 

      ALLMEAT SEAPLANT MONOPOLY DR1TSFAT DR1TSODI G_NWHL EMPTYCAL10;

run;

* This proc means is for verification and is not necessary;
* uncomment next four lines if more information is desired;
*proc means data=SIM_DATA n nmiss min max mean stddev;
*    var DR1TKCAL V_TOTAL V_DRKGR LEGUMES F_TOTAL WHOLEFRT G_WHL D_TOTAL 

      ALLMEAT SEAPLANT MONOPOLY DR1TSFAT DR1TSODI G_NWHL EMPTYCAL10;
*  title2 "Distributions of Simulated Data";
*run;
* This proc print is for verification and is not necessary;
* uncomment next three lines if more information is desired;
*proc print data= SIM_DATA (obs=20);
*  title2 "Listing of 20 Records from Simulated Data";
*run;
/*Section (III): Allocation of beans and peas (legumes) and application of the HEI-2010 scoring algorithm.*/
/*Step 1: Run the macro to properly allocate the intakes of the MPED variable Legumes (LEGUMES) to either Total Protein Foods and Seafood and Plant Proteins (ALLMEAT and SEAPLANT) or Total Vegetables and Greens and Beans (V_TOTAL and V_DRKGR).  The four resulting variables from this step, named LEGUME_ADDED_V_TOTAL, LEGUME_ADDED_BEANGRN, LEGUME_ADDED_ALLMEAT, and LEGUME_ADDED_SEAPLANT are all used in the HEI scoring macro in Step 2*/
%LEG2010A (indat=SIM_DATA,

           kcal= DR1TKCAL,

           allmeat=ALLMEAT,


   seaplant=SEAPLANT,


   v_total=V_TOTAL,


   v_drkgr=V_DRKGR,


   legumes=LEGUMES,


   outdat=LSIM_DATA);

/*Step 2: Apply the HEI-2010 scoring macro. */
%HEI2010 (indat=LSIM_DATA,

          kcal= DR1TKCAL,


  lv_total=LEGUME_ADDED_V_TOTAL,


  lbeangrn=LEGUME_ADDED_BEANGRN,


  f_total=F_TOTAL,


  wholefrt=WHOLEFRT,


  g_whl=G_WHL,


  d_total=D_TOTAL,

       lallmeat=LEGUME_ADDED_ALLMEAT,


  lseaplant=LEGUME_ADDED_SEAPLANT,


  monopoly=MONOPOLY,


  sfat=DR1TSFAT,


  sodi=DR1TSODI,


  g_nwhl=G_NWHL,


  emptycal10=EMPTYCAL10,


  outdat=AFTERMAC);

run;

* This proc means is for verification and is not necessary;
* uncomment next four lines if more information is desired;
*proc means data=AFTERMAC n nmiss min max mean stddev;
*  var legume_added_v_total legume_added_BEANGRN legume_added_allmeat legume_added_SEAPLANT 

    vegden grbnden frtden whfrden wgrnden dairyden

     meatden seaplden faratio sodden rgden sofa_perc;
*  title2 'after legume allocation and hei 2010 scoring macro';
*run;
/*Section (IV): Calculation of HEI-2010 component and total scores and their confidence intervals.*/
/*Step 1.  This step uses univariate and means procedures to compute total and component scores and their standard errors.  */
proc univariate data=AFTERMAC noprint;

  var HEIX1_TOTALVEG HEIX2_GREEN_AND_BEAN HEIX3_TOTALFRUIT HEIX4_WHOLEFRUIT HEIX5_WHOLEGRAIN HEIX6_TOTALDAIRY 

    HEIX7_TOTPROT HEIX8_SEAPLANT_PROT HEIX9_FATTYACID HEIX10_SODIUM HEIX11_REFINEDGRAIN HEIX12_SOFAAS HEI2010_TOTAL_SCORE;

  output out=ci pctlpts=2.5 97.5 pctlpre=h1_ h2_ h3_ h4_ h5_ h6_ h7_ h8_ h9_ h10_ h11_ h12_ totscore_;

run;

proc means data=AFTERMAC noprint;

  var HEIX1_TOTALVEG HEIX2_GREEN_AND_BEAN HEIX3_TOTALFRUIT HEIX4_WHOLEFRUIT HEIX5_WHOLEGRAIN HEIX6_TOTALDAIRY 

    HEIX7_TOTPROT HEIX8_SEAPLANT_PROT HEIX9_FATTYACID HEIX10_SODIUM HEIX11_REFINEDGRAIN HEIX12_SOFAAS HEI2010_TOTAL_SCORE;

  output out=stat min=h1_min h2_min h3_min h4_min h5_min h6_min h7_min h8_min h9_min h10_min h11_min h12_min totscore_min

   max=h1_max h2_max h3_max h4_max h5_max h6_max h7_max h8_max h9_max h10_max h11_max h12_max totscore_max

    mean=h1_mean h2_mean h3_mean h4_mean h5_mean h6_mean h7_mean h8_mean h9_mean h10_mean h11_mean h12_mean totscore_mean

     stddev= h1_stddev h2_stddev h3_stddev h4_stddev h5_stddev h6_stddev h7_stddev h8_stddev h9_stddev h10_stddev h11_stddev h12_stddev totscore_stddev;  

run;

/*Step 2.  This step prepares the results for display*/
data ALL;

  merge CI STAT;

run;

data RESULT (keep=score slabel min max mean stderr lowerci upperci);

  set ALL;

  score='HEIx1          ';

  slabel='HEI-2010 COMPONENT 1 TOTAL VEGETABLES';

  min=h1_min;

  max=h1_max;

  mean=h1_mean;

  stderr=h1_stddev;

  lowerci=h1_2_5;

  upperci=h1_97_5;

  output RESULT;

  score='HEIx2';

  slabel='HEI-2010 COMPONENT 2 GREENS AND BEANS';

  min=h2_min;

  max=h2_max;

  mean=h2_mean;

  stderr=h2_stddev;

  lowerci=h2_2_5;

  upperci=h2_97_5;

  output RESULT;

  score='HEIx3';

  slabel='HEI-2010 COMPONENT 3 TOTAL FRUIT';

  min=h3_min;

  max=h3_max;

  mean=h3_mean;

  stderr=h3_stddev;

  lowerci=h3_2_5;

  upperci=h3_97_5;

  output RESULT;

  score='HEIx4';

  slabel='HEI-2010 COMPONENT 4 WHOLE FRUIT';

  min=h4_min;

  max=h4_max;

  mean=h4_mean;

  stderr=h4_stddev;

  lowerci=h4_2_5;

  upperci=h4_97_5;

  output RESULT;

  score='HEIx5';

  slabel='HEI-2010 COMPONENT 5 WHOLE GRAINS';

  min=h5_min;

  max=h5_max;

  mean=h5_mean;

  stderr=h5_stddev;

  lowerci=h5_2_5;

  upperci=h5_97_5;

  output RESULT;

  score='HEIx6';

  slabel='HEI-2010 COMPONENT 6 DAIRY';

  min=h6_min;

  max=h6_max;

  mean=h6_mean;

  stderr=h6_stddev;

  lowerci=h6_2_5;

  upperci=h6_97_5;

  output RESULT;

  score='HEIx7';

  slabel='HEI-2010 COMPONENT 7 TOTAL PROTEIN FOODS';

  min=h7_min;

  max=h7_max;

  mean=h7_mean;

  stderr=h7_stddev;

  lowerci=h7_2_5;

  upperci=h7_97_5;

  output RESULT;

  score='HEIx8';

  slabel='HEI-2010 COMPONENT 8 SEAFOOD AND PLANT PROTEIN';

  min=h8_min;

  max=h8_max;

  mean=h8_mean;

  stderr=h8_stddev;

  lowerci=h8_2_5;

  upperci=h8_97_5;

  output RESULT;

  score='HEIx9';

  slabel='HEI-2010 COMPONENT 9 FATTY ACID RATIO';

  min=h9_min;

  max=h9_max;

  mean=h9_mean;

  stderr=h9_stddev;

  lowerci=h9_2_5;

  upperci=h9_97_5;

  output RESULT;

  score='HEIx10';

  slabel='HEI-2010 COMPONENT 10 SODIUM';

  min=h10_min;

  max=h10_max;

  mean=h10_mean;

  stderr=h10_stddev;

  lowerci=h10_2_5;

  upperci=h10_97_5;

  output RESULT;

  score='HEIx11';

  slabel='HEI-2010 COMPONENT 11 REFINED GRAINS';

  min=h11_min;

  max=h11_max;

  mean=h11_mean;

  stderr=h11_stddev;

  lowerci=h11_2_5;

  upperci=h11_97_5;

  output RESULT;

  score='HEIx12';

  slabel='HEI-2010 COMPONENT 12 SOFAAS CALORIES';

  min=h12_min;

  max=h12_max;

  mean=h12_mean;

  stderr=h12_stddev;

  lowerci=h12_2_5;

  upperci=h12_97_5;

  output RESULT;

  score='TOTAL HEI 2010';

  slabel='TOTAL HEI-2010 SCORE';

  min=totscore_min;

  max=totscore_max;

  mean=totscore_mean;

  stderr=totscore_stddev;

  lowerci=totscore_2_5;

  upperci=totscore_97_5;

  output RESULT;

run;

/*Step 3 - This step displays the results*/
proc print data=RESULT;

  id score;

  var slabel min max mean stderr lowerci upperci;

  title2 'complex survey design population method - mean and confidence interval of HEI-2010 using NH 03/04 data';

run;

/*Step 4 - This step provides an option to export the results into a CSV file that can be opened in Excel.*/
proc export data=RESULT

  file=RES

  dbms=csv

  replace;

run;

